



## Integral de Linha

### Introdução

Este procedimento calcula a integral de linha de um campo vetorial  $F$  ao longo de uma curva parametrizada.


```
> lineint:=proc(F,r,t1,t2)  
local Ft, v;  
v:=diff(r,t);  
Ft:=subs(x=r[1],y=r[2],F);  
int(Ft[1]*v[1]+Ft[2]*v[2],  
t=t1..t2);  
end;
```

```
> H:=[-y/(x^2+y^2),x/(x^2+y^2)];
```

$$H = \left[ -\frac{y}{x^2 + y^2}, \frac{x}{x^2 + y^2} \right]$$

```
> with(plots):
```

```
> FP:=fieldplot(H,x=-5..5,y=-5..5,  
grid=[12,12],color=blue,thickness=3,scaling=constrained):  
FP;
```



> **lineint(H,[cos(t),sin(t)],0,2\*Pi);**

2 π

## Um exemplo útil

Este procedimento plota o segmento de reta determinada por dois pontos e calcula a integral de linha ao longo de cada segmento:

```
> PQint:=proc(F,P,Q)
local xt, yt, Ft;
xt:=P[1]+t*(Q[1]-P[1]);
yt:=P[2]+t*(Q[2]-P[2]);
Ft:=subs(x=xt,y=yt,F);
evalf(int(Ft[1]*(Q[1]-P[1])+
Ft[2]*(Q[2]-P[2]),
t=0..1)):
end:
```

```
> A:=[3,4]:
B:=[-5,-2]:
C:=[2,1]:
q1:=PQint(H,A,B);
q2:=PQint(H,B,C);
q3:=PQint(H,C,A);
```

*q1* := 2.594803813


```
q2 := -3.058451422
```

```
q3 := .4636476085
```

```
> mysum:=q1+q2+q3;
```

```
mysum := -5 10-9
```

```
> display([FP,polygonplot([A,B,C],  
color=yellow)],  
title=convert(mysum,string));
```



```
>
```